Computation of the
1982 Kimberly-Penman
and the
Jensen-Haise
Evapotranspiration Equations
as Applied in the
U.S. Bureau of Reclamation's
Pacific Northwest AgriMet Program

February 1994

Prepared by

Doug Dockter
U. S. Bureau of Reclamation
Pacific Northwest Region
Water Conservation Center

Revised January, 2008 Peter L. Palmer AgriMet Program Manager



Table of Contents

	Page
INTRODUCTION	1
THE 1982 KIMBERLY-PENMAN EQUATION	2
Net Radiation (the heat function)	
The slope of the saturation vapor pressure-temperature curve,	
The psychrometric constant	
The net radiation energy	
The soil heat flux	
Advective Energy Transfer (the wind function)	
The constant of proportionality The dimensionless wind function	
The mean daily vapor pressure deficit	
The Conversion of the Kimberly-Penman ET _r in MJ/m ² /d to inches/d	
THE JENSEN-HAISE EQUATION	11
The temperature coefficient	
Temperature, $(T_{mean} - T_x)$	
The global solar radiation, R	
The Conversion of Jensen-Haise ET, in cal/cm²/d to inches/d	13
APPENDIX A - Subroutine ETPCOMP	14
APPENDIX B - Program JENSEN	22
APPENDIX C - Data File CLIMAT.DAT	24
REFERENCES	26

INTRODUCTION

AgriMet is a cooperative agricultural weather station network, managed and operated by the U.S. Bureau of Reclamation, dedicated to agricultural crop water use modeling. The program supports improved irrigation management and scheduling using the Kimberly-Penman combination equation to model and calculate daily water use for a variety of agricultural crops grown in the Pacific Northwest. The Kimberly-Penman equation was determined to be applicable to the area served by AgriMet based on research performed by Reclamation and the Agricultural Research Service (ARS) in Kimberly, Idaho during the early 1970's. This report documents the computation of the 1982 Kimberly-Penman and the Jensen-Haise equations as they are applied in the U.S. Bureau of Reclamation's Pacific Northwest AgriMet Program. (Note: the Jensen-Haise procedure was discontinued in 1994).

AgriMet uses the 1982 Kimberly-Penman equation for computing reference ET, adapted by Dr. James L. Wright of the USDA Agricultural Research Service through his research performed in Kimberly, Idaho (Jensen et al., 1990). This procedure requires several meteorological inputs for modeling ET, including maximum and minimum daily air temperatures, relative humidity, daily solar radiation, and daily wind run. All of these parameters are collected by the AgriMet system. ETr calculated by AgriMet is based on a daily (not hourly) time step. The 1982 Kimberly-Penman model uses alfalfa as the reference crop with reference conditions defined as a well-watered alfalfa crop with 30 to 50 cm of top growth.

Because of the variability of ET rates from crop to crop and the complexity of modeling ET, the accepted standard for deriving crop specific ET (ETc) is to model ET for a reference crop, such as alfalfa (ETr), and then apply this reference ETr value to specific crops through the use of crop coefficients (Kc). These crop coefficients are unique to the reference crop and the individual specific crop, and they vary through time with the growth stage of the plant. These crop coefficients typically are expressed as percent of water use compared to the reference crop. Most of the crop coefficients used by AgriMet were developed through field research conducted by the USDA Agricultural Research Service at Kimberly, Idaho.

The Kimberly-Penman and Jensen-Haise computations are performed in the FORTRAN subroutine program ETPCOMP.FOR (ETPCOMP) (see Appendix A). The weather data required for the computations are supplied via satellite from AgriMet weather stations located throughout the Pacific Northwest. The AgriMet weather data inputs for temperature, wind run, and solar radiation are converted from English to metric SI units in ETPCOMP to accommodate the use of the Kimberly-Penman equation with a minimum of conversion constants.

Daily Jensen-Haise values are calculated and reported within the AgriMet database prior to 1995 as a reference for users and are not used in the AgriMet crop models.

THE 1982 KIMBERLY-PENMAN EQUATION

Since it was first derived by H. L. Penman in 1948, the original Penman equation has been modified a number of times. The AgriMet program uses the 1982 Kimberly-Penman equation adapted by Dr. James L. Wright of the ARS through his research performed in Kimberly, Idaho. The 1982 Kimberly-Penman uses alfalfa as the reference crop with reference conditions established as well-watered with 30 to 50 cm of top growth. The Penman equation used under AgriMet program crop modeling is as follows:

$$\lambda ET_r = \frac{\Delta}{\Delta + \gamma} (R_n - G) + \frac{\gamma}{\Delta + \gamma} 6.43 W_f (e_s - e_a)$$

with units in MJ/m²/d. ETPCOMP will not compute values less than zero.

(Note: (R_n-G) differs from the (R_n+G) term seen in some publications, due to the fact that G is calculated here with an opposite sign convention.)

The expression for the Penman equation appears in ETPCOMP under the variable name ETP LAMB.

The Penman equation is known as a "combination equation" because it combines **net radiation** (*the heat function*) and **advective energy transfer** (*the wind function*) into one energy balance equation.

The sum of the two terms

$$\frac{\Delta}{\Delta + \gamma}$$
 and $\frac{\gamma}{\Delta + \gamma}$

is equal to one. These terms represent weighting factors that assess the relative effects of the heat function and the wind function on evaporation. These weightings are approximately 75% *heat function* and 25% *wind function* for the general climatic conditions during the growing season in the Pacific Northwest.

The weighting factors for the *heat function* and the *wind function* appear in ETPCOMP under the variable names FACT1 and FACT2 respectively.

Net Radiation (the heat function)

The *heat function* portion of the equation consists of:

$$\frac{\Delta}{\Delta+\gamma}(R_n-G)$$

where:

 Δ is the slope of the saturation vapor pressure-temperature curve

 γ is the psychrometric constant

R_n is the net radiation

G is the soil heat flux

The slope of the saturation vapor pressure-temperature curve, \triangle , represents the change in saturation vapor pressure with respect to temperature (kPa/°C). It is derived by differentiating the following expression, presented by J. F. Bosen in 1960, that estimates the saturation vapor pressure over water (kPa) (for -51 °C < T < 54 °C):

$$e^{-} = 3.38639[(0.00738T + 0.8072)^{8} - 0.000019/1.8T + 48/+0.001316]$$

By making the substitution, U = 0.00738T + 0.8072, and taking the derivative of the above equation with respect to T,

$$\Delta = 3.38639(0.05904(0.00738T + 0.8072)^7 - 0.0000342)$$

Further simplifying this, and substituting the mean daily temperature for the temperature variable T, the equation becomes

$$\Delta = 0.200(0.00738 T_{mean} + 0.8072)^7 - 0.000116$$

(Note: this equation is valid for $T \ge -23$ °C, however increasing errors occur for temperatures below about -6 °C, due to changes in the relationship between vapor pressure and temperature under these conditions.)

The expression for \triangle appears in ETPCOMP under the variable name DD and uses the mean daily temperature in °C (converted from AgriMet data in °F).

The psychrometric constant, \forall , (kPa/°C) represents a balance between the sensible heat gained from air flowing past a wet bulb thermometer and the sensible heat transformed into latent heat (Brunt, 1952). It varies directly with air pressure and is calculated as follows:

$$\gamma = \frac{c_p P}{0.622 \lambda}$$

where:

 c_p is the specific heat of air at constant pressure and is equal to 0.001005 MJ/kg/°C.

P is the estimated mean atmospheric pressure in kPa.

0.622 is the ratio of the molecular weights of water vapor to air.

 λ is the latent heat of vaporization of water in MJ/kg calculated at the mean daily temperature.

P, the mean atmospheric pressure in kPa, is found by applying an equation derived from the ideal gas law that is used for estimating pressure variation with altitude in the Troposphere:

$$P = P_o \left[\frac{(T_o - \alpha E)}{T_o} \right]^{g/(\alpha R)}$$

where:

P_o is the standard atmospheric pressure and is equal to 101.3 kPa.

T_o is the absolute temperature of a standard atmosphere and is equal to 288 °K.

E is the elevation of the weather station in meters above mean sea level and is stored under the variable name ELEV in the data file CLIMAT.DAT (see Appendix C).

 α is the lapse rate, the rate at which the temperature decreases linearly with increasing elevation and is equal to 0.0065 $^{\circ}\text{K/m}.$

g is the acceleration of gravity and is equal to 9.8067 m/sec².

R is the gas constant and is equal to 287 J/kg/°K for dry air.

Substituting these values into the pressure equation above, the expression becomes:

$$P = 101.3\left(\frac{288 - 0.0065E}{288}\right)^{5.257}$$

 λ , the latent heat of vaporization of water in MJ/kg, is calculated at the mean daily temperature in °C (converted from AgriMet data in °F) as follows:

$$\lambda = 2.501 - 0.002361_{T_{mean}}$$

The expressions for γ , P, and λ appear in ETPCOMP under the variable names GG, PP, and PLAMBDA respectively.

The net radiation energy, R_n , (MJ/m²/d) received by a well-watered irrigated crop has been shown to have a close relationship to the ET rate, and is calculated as follows:

$$R_n = (1 - \alpha)R_s - R_b$$

where:

 α is the reflected shortwave (solar) radiation, albedo, and is calculated as:

$$\alpha = 0.29 + 0.06\sin(x + 97.92)$$

(<u>Note</u>: The expression relates the angle of the sun to the time of year. This expression differs slightly from other published versions to accommodate the use of the day number (x) in the year (Dr. James L. Wright, personal contact February, 1994). $(1-\alpha)$ represents the percentage of solar radiation received by the crop-soil surface).

R_s is the measured solar (shortwave) radiation value in MJ/m²/d (converted from AgriMet weather station pyranometer data in cal/cm²/d).

R_b is the estimated outgoing thermal (longwave) radiation in MJ/m²/d and is calculated as follows:

$$R_b = R_{bo}(a\frac{R_s}{R_{so}} + b)$$

where:

The term a(R_s/R_{so})+b is a factor that adjusts the net clear day outgoing longwave radiation for cloud conditions.

a and b are empirical constants determined from research at Kimberly, Idaho.

$$a = 1.126$$
 and $b = -0.07$ for $R_s/R_{so} > 0.70$

$$a = 1.017$$
 and $b = -0.06$ for $R_s/R_{so} \le 0.70$

R_s is the measured shortwave radiation value in MJ/m²/d (converted from AgriMet weather station pyranometer data in cal/cm²/d).

R_{so} is the measured clear-day shortwave radiation in cal/cm²/d. It is represented by a polynomial equation

$$\mathbf{R}_{so} = \mathbf{C}_1 + \mathbf{C}_2 \mathbf{x} + \mathbf{C}_3 \mathbf{x}^2 + \mathbf{C}_4 \mathbf{x}^3 + \mathbf{C}_5 \mathbf{x}^4$$

where:

x is the day number in the year and the coefficients are unique for each AgriMet weather station site and are read in from CLIMAT.DAT (see Appendix C). This calculation is performed in English units and then converted to MJ/m²/d since the coefficients at each AgriMet site were originally determined in English units.

R_{bo} is the theoretical net outgoing longwave radiation expected to occur on a clear day in

$$R_{bo} = (a_1 + b_1 \sqrt{e_a}) 4.903 \times 10^{-9} (\frac{T_{\text{max}}^4 + T_{\text{min}}^4}{2})$$

MJ/m²/d and is calculated as follows:

where:

The term $a_1+b_1\sqrt(e_a)$ is the net emittance expression and represents the difference between the longwave emittance for the reference crop and the effective emittance for the atmosphere.

a₁ and b₂ are empirical constants determined from research at Kimberly, Idaho. They account for the fact that water vapor, smoke and dust particles absorb shortwave radiation from the sun and emit it as longwave

radiation according to their characteristic temperatures.

a₁ is calculated as follows:

$$a_1 = 0.26 + 0.1e^{-((.0154(x-177))^2)}$$

(Note: This expression differs slightly from other published versions to accommodate the use of the day number (x) in the year (Dr. James L. Wright, personal contact February, 1994).

$$b_1 = -0.139$$

e_a is the saturation vapor pressure (kPa) obtained at the mean daily dewpoint temperature using the Bosen equation,

$$e_a = 3.38639[(0.00738T_{dew} + 0.8072)^8 - 0.000019|1.8T_{dew} + 48| + 0.001316]$$

 T_{dew} is the mean daily dewpoint in $^{\circ}C$ (converted from the AgriMet data in $^{\circ}F$).

 4.903×10^{-9} in MJ/m²/d/°K⁴ is the Stefan-Boltzman constant. It is an empirical coefficient representing the proportionality of the emissivity of a body to temperature.

 T_{max} and T_{min} are the daily maximum and minimum temperatures in ${}^{\circ}K$ (converted from the AgriMet data in ${}^{\circ}F$).

The expressions for R_n , R_s , R_b and R_{bo} appear in ETPCOMP under the variable names RN, SOLAR, RB, and RBO respectively.

The expression for R_{so} appears in ETPCOMP under the variable name CLRAD, with the value converted to $MJ/m^2/d$ and stored under the variable name RSO.

The term $a_1 + b_1 \sqrt{(e_a)}$ appears in ETPCOMP under the variable name EMIT.

The expression for e_a appears in ETPCOMP under the variable name VAPDP.

The soil heat flux, G, (MJ/m²/d) is determined from an empirical equation developed through research conducted on soils at Kimberly, Idaho. It is based on an assumed 50% available moisture content in the soil. This translates to about 24% soil water content by volume or 16% soil water content by weight for a silt-loam soil. The soil heat flux is calculated as follows:

$$G = 0.377(T_{mean} - T_{nr})$$

where:

 T_{mean} is the mean daily temperature in °C (converted from AgriMet data in °F).

 T_{pr} is the average of the previous three days mean temperatures $^{\circ}C$ (converted from AgriMet data in $^{\circ}F$).

(Note: This expression differs slightly from other published versions to accommodate the use of metric units (Dr. James L. Wright, personal contact February, 1994). Under this sign convention, G is positive when the soil is warming. G is subtracted from R_n in the heat function portion of the Penman equation since, if the soil is warming, then part of the net radiation is being used to warm the soil and less is available to evaporate water).

The expressions for G, T_{mean} and T_{pr} appear in ETPCOMP under the variable names HTFLUX, TMEAN, and TPREAV respectively.

Advective Energy Transfer (the wind function)

The wind function portion of the equation consists of:

$$\frac{\gamma}{\Delta + \gamma} 6.43 W_f(e_s - e_a)$$

where:

 $\gamma/(\gamma+\Delta)$ is the wind function weighting factor made up of γ , the psychrometric constant and Δ , the slope of the saturation vapor pressure-temperature curve as described above.

6.43 is the constant of proportionality in MJ/m²/d/kPa.

W_f is the dimensionless wind function.

(e_s - e_a) is the mean daily vapor pressure deficit in kPa.

The constant of proportionality, 6.43 in MJ/m²/d/kPa, relates the dimensionless wind function to the vapor pressure deficit in the equation.

The dimensionless wind function characterizes the effects of wind in the advection (motion of the atmosphere) of sensible heat for ET and is calculated as follows:

$$\mathbf{W}_{\mathrm{f}} = \mathbf{a}_{\mathrm{w}} + \mathbf{b}_{\mathrm{w}} \mathbf{U}_{\mathrm{2}}$$

where:

U₂ is the wind speed in kilometers per day (converted from AgriMet data in miles per day).

a_w and b_w are empirical coefficients that lessen the effect of the wind function during the early part of the year. These are based on the assumption that the surrounding rangeland is wetter early in the season and has less advection of sensible heat than occurs later in the year when surrounding areas are drier. These coefficients assume an anemometer height of 2 meters.

 $a_{_{\rm w}}$ and $b_{_{\rm w}}$ are calculated as follows with x representing the day number in the calendar year:

$$a_w = 0.4 + 1.4 e^{-((\frac{x-173}{58})^2)}$$

$$b_w = 0.007 + 0.004 e^{-\left(\left(\frac{x-243}{80}\right)^2\right)}$$

The expressions for W_r , a_w , b_w , and U_2 appear in ETPCOMP under the variable names WF, AW, BW, and U2 respectively.

The mean daily vapor pressure deficit is expressed by the quantity (e_s - e_a) in kPa.

where:

The term e_s is the average saturation vapor pressure (kPa) calculated from the saturation vapor pressures determined at the maximum and minimum daily temperatures as follows:

$$e_s = \frac{e_{Tmax} + e_{Tmin}}{2}$$

where:

 $e_{_{Tmax}}$ and $e_{_{Tmin}}$ are in kPa and are determined from the Bosen equation at the maximum and minimum daily air temperatures, $T_{_{max}}$ and $T_{_{min}}$ in $^{\circ}C$ (converted from AgriMet data in $^{\circ}F$) as follows:

$$e_{_{Tmax}} = 3.38639[(0.00738T_{_{max}} + 0.8072)^8 - 0.000019\big|1.8T_{_{max}} + 48\big| + 0.001316]$$

$$e_{_{Tmin}} = 3.38639[(0.00738T_{_{min}} + 0.8072)^{^{8}} - 0.000019 \big| 1.8T_{_{min}} + 48 \big| + 0.001316]$$

 e_a is the saturation vapor pressure (kPa) determined from the Bosen equation at the mean daily dewpoint temperature, T_{dew} , in $^{\circ}C$ (converted from AgriMet data in $^{\circ}F$) as follows:

$$e_{_a} = 3.38639[(0.00738T_{_{dew}} + 0.8072)^8 - 0.000019 \big| 1.8T_{_{dew}} + 48 \big| + 0.001316]$$

The expressions for e_{Tmax} , e_{Tmin} , and e_{a} appear in ETPCOMP under the variable names VAPMX, VAPMN, and VAPDP respectively.

The Conversion of the Kimberly-Penman ET_r in $MJ/m^2/d$ to inches/d

Conversion of the units to inches per day is accomplished by multiplying the results of the equation in $MJ/m^2/d$ by the following:

$$\frac{(1x10^{3}\frac{cm^{3}}{kg})(1x10^{-6}\frac{m^{3}}{cm^{3}})(39.37\frac{inches}{m})}{\lambda \frac{MJ}{kg}}$$

where the latent heat of water in MJ/kg is $\lambda = 2.501 - 0.002361T_{mean}$.

THE JENSEN-HAISE EQUATION

The Jensen-Haise equation (cal/cm²/d) represents a temperature-radiation method of calculating a daily ET_r. Although the AgriMet program uses the 1982 Kimberly-Penman for crop modeling, it also calculates Jensen-Haise values as an additional reference (also using alfalfa as the reference crop with conditions established as well watered with 30 to 50 cm of top growth). English units are used in the calculation since the coefficients have been determined for each individual weather station site in English units. The Jensen-Haise equation is as follows:

$$\lambda ET_r = C_T (T_{mean} - T_X) R_s$$

where:

 $C_{\scriptscriptstyle T}$ is the temperature coefficient.

 T_{mean} is the daily mean temperature in °F.

T_x is the intercept of the temperature axis as calculated for each AgriMet weather station.

R_s is the measured global solar radiation in cal/cm²/day

The product of $C_T(T_{mean}-T_x)$ represents a weighting function on R_s .

The expression for the Jensen-Haise equation appears in ETPCOMP under the variable name ETPJ_VALUE.

The temperature coefficient, C_T , is an empirical coefficient based on temperature and atmospheric pressure. It relates temperature and radiation to ET. C_T is unique for each AgriMet weather station site and is calculated using the FORTRAN program JENSEN.FOR (JENSEN)(see Appendix B). The values for each station are stored under the variable name CSUBT in a file that is accessed by ETPCOMP called CLIMAT.DAT (see Appendix C). C_T is calculated as follows:

$$C_T = \frac{1}{C_1 + C_2 C_H}$$

where:

 C_1 , C_2 , and C_H are empirically derived correlation coefficients as follows:

where:

ELEV is the elevation of the AgriMet weather station in feet above mean sea level (from prompted input to JENSEN).

$$C_1 = 68 - 3.6 \frac{ELEV}{1000}$$

$$C_2 = 13^{\circ}F$$

$$C_H = \frac{50 \text{ milibars}}{e_2 - e_1}$$

where:

 $e_{_{2}}$ and $e_{_{1}}$ are vapor pressures determined at $T_{_{max}}$ and $T_{_{min}}$ and are calculated as follows ($e_{_{1}}$ is the same equation except for the substitution of $T_{_{min}}$ for $T_{_{max}}$):

$$e_2 = 6.105 \, e^{(25.22 \frac{(\frac{5}{9} (T_{\text{max}} \cdot 32) + 273) \cdot 273}{\frac{5}{9} (T_{\text{max}} \cdot 32) + 273} - 5.3I \ln(\frac{\frac{5}{9} (T_{\text{max}} \cdot 32) + 273}{273}))}$$

where:

 $T_{\mbox{\tiny max}}$ and $T_{\mbox{\tiny min}}$ are the long term mean maximum and mean minimum air temperatures respectively for the month of the year with the highest mean air temperatures in °F (from prompted input to JENSEN).

Temperature, (T_{mean} - T_x), is the difference between T_{mean} , the daily mean temperature (from AgriMet data in °F), and T_x , the intercept of the temperature axis. T_x is unique for each AgriMet weather station site and is calculated in JENSEN. The values for each station are stored under the variable name TSUBX in CLIMAT.DAT. T_x is calculated as follows:

$$T_x = 27.5 - 0.25(e_2 - e_1) - \frac{ELEV}{1000}$$

where:

ELEV is the elevation of the AgriMet weather station in feet above mean sea level (from prompted input to JENSEN).

The global solar radiation, R_s , is the measured solar radiation value in cal/cm²/d from AgriMet weather station pyranometer data. It appears in ETPCOMP under the variable name SR_VALUE.

The Conversion of Jensen-Haise ET_r in cal/cm²/d to inches/d

Conversion of the units to inches/d is accomplished by multiplying the results of the equation in cal/cm²/d by

$$\frac{0.3937 \frac{inches}{cm}}{(\lambda \frac{cal}{g})(1 \frac{g}{cm^{3}} of water)}$$

where:

The latent heat of water in cal/g is $\lambda = 595$ - $0.51T_{\text{mean}}$. T_{mean} is in °C (converted from AgriMet data in °F) for this calculation.

APPENDIX A - Subroutine ETPCOMP

	xxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxx
C	$\times \times $
	ROUTINE TITLE: SUBROUTINE ETPCOMP (FEB 1994)
	PREVIOUS VERSION SAVED AS ETPCOMP.OLD
Z	PURPOSE: COMPUTES EVAPOTRANSPIRATION RATE USING THE 1982
	KIMBERLY-PENMAN EQUATION AND THE JENSEN-HAISE
2	EOUATION.
	AUTHOR: JOHN HOCKETT - USBR
C	MODIFICATION - KEEFER - NOVEMBER 87
	FOUND THE FILE CLIMAT.DAT AND MOVED IT TO
	THE DMS SYSTEM FROM OLD BMHS. MODIFIED THE
C	LIST-DIRECTED READ FOR THE 5 CONSTANTS SO
2	THE FILE WOULD OPEN AND READ CORRECTLY.
C	
	KEEFER - MAR 88
C	
C	ADDED LIMITS CHECKS FOR ALL INPUT VARIABLES
	POWERS - JAN 89
	MODIFIED CANAL AND CANAL CALCULATION DECERDING
	MODIFIED GAMM1 AND GAMM2 CALCULATION PROCEDURE
	BASED ON UPDATED ELEVATION FUNCTION AS PER
	JIM WRIGHT (IN ADDITION, THIS VERSION SET UP
C	WITHIN DUA12:[APOWERS.ET] INDEPENDENT OF DMS TO BE AUTOMATICALLY RUN USING ARCHIVE DATA THROUGH
2	DUA12: [APOWERS.ET] AUTOACMET.COM
Z	GAMM1 AND GAMM2 ARE NOW CALLED FACT1 AND FACT2
2	(DOCKTER - FEB 1994)
2	(BOOKIEK TEB 1991)
Ž	BROWER - APR 89
2	21.01.21. 121. 05
2	MODIFIED TO CALCULATE THE JENSEN ETP VALUE,
	CALLED ETPJ VALUE.
	POWERS - APR 89
;	
	PROGRAMS ETPCOMP, ACMET, ACMETALL, AUTOACMET &
	CLIMAT.DAT MODIFIED AND MOVED TO OPERATE
	FROM DUA12:[AGRIMET.ET]
•	POWERS - MAY 89

С С С С С C C С С С С С С С С С C С С С С С C C С С С C C C С С С C C С C С C C C С С С

С

С

С

ADDED "NO VALUE" CHECKS FOR OB_MIN AND OB_MAX AND SET UP <0.00 CHECK FOR ETP_VALUE AND ETPJ_VALUE.

POWERS - SEPT 89

SUBSTITUTED MEAN DAILY TEMPERATURE (MM_SAVE)
TO BE USED INSTEAD OF OB_MIN AND OB_MAX IN
HEAT FLUX (PREAV) AND NET RADIATION (TMPKE)
CALCULATIONS.

TMPKE IS NO LONGER USED BUT IS TAKEN CARE OF IN CALCULATION OF RBO (DDOCKTER - FEB 1994).

DOCKTER - FEB 94

- CHANGED FORMULAS SO THAT THEY ARE MORE LITERAL AS SEEN IN TECHNICAL PUBLICATIONS. MAKES CODE EASIER TO FOLLOW.
- CHANGED VAPOR PRESSURE, RBO, AND WIND EQUATIONS TO MAKE COMPATIBLE WITH 1982 KIMBERLY-PENMAN METHODS (SEE FOLLOWING COMMENTS).
- ADDED FORMULAS FOR COMPUTING A1, A, B, AND ALBEDO TO MAKE COMPATIBLE WITH 1982 KIMBERLY-PENMAN METHODS.
- ADDED A SECTION TO CONVERT OB_MEAN, OB_MAX, OB_MIN TP_VALUE FROM FAHRENHEIT TO CELSIUS AND KELVIN, SR_VALUE FROM CAL/SQCM/DAY TO MJ/SQ M/DAY, AND WR VALUE FROM MILES/DAY TO KM/DAY.
- CHANGED THE COMPUTATION OF EJ SO IT USES DAILY MEAN TEMP INSTEAD OF DAILY AVERAGE TEMP
- CHANGED FORMULAS SO LL UNITS ARE IN MJ/kg/DEG C,
 DD UNITS ARE IN kPA/DEGREE C, PP UNITS ARE IN kPa,
 AND GG UNITS ARE IN kPa/DEGREE C, G UNITS ARE IN
 MJ/SQ M/D, MM SAVE UNITS ARE IN DEGREE C
- CHANGED SATURATION VAPOR PRESSURE FORMULAS SO THAT VAPMA, VAPMI, AND VAPDP ARE CALCULATED USING THE BOSEN EQUATION RATHER THAN THE OLD POLYNOMIAL EQU. UNITS ARE NOW IN kPA INSTEAD OF MILIBARS.
- CHANGED THE COMPUTATION OF RBO SO THAT IT USES

 (MAXK**4+MINK**4)/2 INSTEAD OF OLD TEMPKE VARIABLE

 CHANGED THE FACTOR FOR CONVERTING ET TO INCHES/DAY

 SO IT USES ACTUAL COMPUTED LAMBDA INSTEAD OF AN

 ASSUMED AVERAGE VALUE OF LAMBDA.

SUBROUTINE ETPCOMP (*, MM SAVE, OB MEAN,

- 1 WR VALUE, OB MAX, OB MIN, TP VALUE, SR VALUE, NOVAL,
- 2 ETP VALUE, ETPJ VALUE, MONTH, DAY, STATION, LOGUNIT)

```
С
С
      MM SAVE
                     = PREVIOUS DAYS MEAN TEMP IN F
      OB_MEAN
С
                 = TODAY MEAN TEMP IN F
     WR_VALUE = WIND RUN

OB_MAX = TODAYS MAX TEMP IN F

OB_MIN = TODAYS MIN TEMP IN F

TP_VALUE = TODAYS DEW POINT IN F

SR_VALUE = SOLAR RADIATION

NOVAL = 998877.00

MONTH = MONTH NAME
С
С
С
С
С
С
С
     MONTH
                 = MONTH NAME
С
                = CHARACTER DAY NUMBER
     DAY
     STATION = COLUMBIA BASIN TELETYPE (CBTT) STATION NAME
LOGUNIT = OUTPUT UNIT FOR ERROR MESSAGES
С
С
С
        REAL KMIN, KMAX, JLAMBDA
      REAL*4 C(4), MM SAVE(3), CLDYC(5), NOVAL
      INTEGER*2 KOUNT
      INTEGER*4 IEND(13)
      CHARACTER MONTH*3, DAY*2, MONTHS ALL*48, STATION*12, STN*12
С
      DATA MONTHS ALL/'JAN*FEB*MAR*APR*MAY*JUN*JUL*AUG*
           SEP*OCT*NOV*DEC*'/
      DATA IEND/0,31,59,90,120,151,181,212,243,273,304,
            334,365/
С
С
      С
      BEGIN EXECUTABLE CODE
С
      C
      NOVAL=998877
      ETP VALUE=NOVAL
       ETPJ VALUE=NOVAL
      IF (OB MEAN.EQ.NOVAL.OR.TP VALUE.EQ.NOVAL.OR.
      1
            WR VALUE.EQ.NOVAL.OR.SR VALUE.EQ.NOVAL.OR.OB MIN.
            EQ.NOVAL.OR.OB MAX.EQ.NOVAL) THEN
            RETURN ! NOT ENOUGH DATA TO CALC ETP
      END IF
С
С
      LIMIT WIND RUN TO 150 MILES PER DAY.
C
      IF (WR VALUE.GT.150.0) WR VALUE=150.0
С
С
      READ CONSTANTS FROM CLIMAT.DAT --
С
        STN = WEATHER STATION COLUMBIA BASIN TELETYPE CODE
С
        CLDYC= CONSISTS OF 5 CLEARDAY SOLAR RADIATION COEFFICIENTS
С
        CLDYM= THE CLEAR DAY MINIMUM SOLAR RADIATION
С
        ANEMH= ANEMOMETER HEIGHT - THIS VALUE IS NO LONGER USED IN
С
               THE COMPUTATIONS SINCE ALL ANEMOMETERS IN THE FIELD
С
               HAVE BEEN SET TO APROX. 2 METERS IN HEIGHT AS IS
С
               ASSUMED BY THE 1982 KIMBERLY-PENMAN EOU. CLIMAT.DAT
```

```
С
               SHOWS THESE STATIONS TO BE AT 7FT WHICH IS OUTDATED
С
               INFO AT PRESENT
С
     ELEV = WEATHER STATION ELEVATION IN METERS
С
       CSUBT= TEMPERATURE COEFFICIENT JENSEN-HAISE EQU.
С
        TSUBX= INTERCEPT OF THE TEMPERATURE AXIS FOR JENSEN-HAISE
С
               EOUATION
С
        OPEN (UNIT=76, FILE='DUA12: [AGRIMET.ET] CLIMAT.DAT', TYPE='OLD',
            READONLY, ERR=1000)
30
       READ(76, '(A4, 3F9.0, F10.0, F9.0, F6.0, F4.0, 2F7.0, F8.0)', ERR=1000,
     2 END=1000,
            IOSTAT=ISTAT) STN, CLDYC, CLDYM, ANEMH, ELEV, CSUBT, TSUBX
     IF (STN.NE.STATION) GO TO 30
С
С
        CONVERT WEATHER DATA TO SI UNITS
С
        TMEAN=5.* (OB MEAN-32.)/9.
        TMAX=5.*(OB MAX-32.)/9.
        TMIN=5.*(OB MIN-32.)/9.
        TDEW=5.*(TP VALUE-32.)/9.
        KMAX=TMAX+273.16
        KMIN=TMIN+273.16
        SOLAR=SR VALUE*0.041868
        U2=WR VALUE*1.6093
С
С
        COMPUTE LAMBDA IN CAL/G FOR JENSEN
С
        JLAMBDA=595.-0.51*TMEAN
С
С
        COMPUTE JENSEN ETP - THIS IS IN ENGLISH UNITS BECAUSE
С
          CSUBT AND TSUBX FOR ALL STATIONS (CLIMAT.DAT) HAVE BEEN
С
          CALCULATED IN ENGLISH UNITS. CHANGED FORMULA TO USE
С
          OB MEAN INSTEAD OF CALCULATING AN AVERAGE TEMPERATURE
С
          (FEB 94-DDOCKTER)
С
        ETPJ VALUE=CSUBT*(OB MEAN-TSUBX)*SR VALUE*0.3937/JLAMBDA
С
С
        BEGIN COMPUTATIONS FOR PENMAN ET
С
С
        COMPUTE LAMBDA FOR PENMAN IN MJ/kg, DELTA (DD) IN kPa/deg C,
С
           ATMOS. PRESSURE (PP) IN kPa, GAMMA (GG) IN kPa/deg C,
С
           HEAT AND WIND FUNCTION WEIGHTING FACTORS (FACT1 AND FACT2)
С
           ARE DIMENSIONLESS.
      PLAMBDA=2.501-0.002361*TMEAN
      DD=0.2*(0.00738*TMEAN+0.8072)**7.-0.000116
      PP=101.3*(((288.-0.0065*ELEV)/288.)**5.257)
      GG=0.001005*(PP)/(0.622*PLAMBDA)
      FACT1=DD/(DD+GG)
      FACT2=1.-FACT1
```

```
С
С
      COMPUTE AVERAGE TEMP FOR PRIOR THREE DAYS USING MM
С
      IF PRIOR DAYS ARE NOVAL THEN USE TODAY'S MM
С
      PREAV=0.0
      KOUNT=0
      DO 10 L=1,3
      IF (MM SAVE(L).EQ.NOVAL)GO TO 10
      PREAV=PREAV+MM SAVE(L)
      KOUNT=KOUNT+1
10
      CONTINUE
      IF (KOUNT.EQ.0) THEN
            PREAV=OB MEAN
      ELSE
            PREAV=PREAV/FLOATI (KOUNT)
      END IF
С
С
       CONVERT PREAV TO CELSIUS
С
        TPREAV=5.*(PREAV-32.)/9.
С
С
       FIND HEAT FLUX, HTFLUX
С
      HTFLUX= (TMEAN-TPREAV) *0.377
С
С
        FIND MIN VAPOR PRESSURE
С
      VAPMN=3.38639*((.00738*TMIN+.8072)**8.-.000019*ABS(1.8*TMIN+48.)
      1 + 0.001316)
С
С
     FIND MAX VAPOR PRESSURE
С
      VAPMX=3.38639*((.00738*TMAX+.8072)**8.-.000019*ABS(1.8*TMAX+48.)
      1 + 0.001316
С
С
     FIND DEW POINT VAPOR PRESSURE
С
      VAPDP=3.38639*((.00738*TDEW+.8072)**8.-.000019*ABS(1.8*TDEW+48.)
      1 + 0.001316
С
С
      FIGURE OUT THE DAY NUMBER IN THE CALENDAR YEAR
С
      MO=INDEX (MONTHS ALL, MONTH) /4+1
      READ (UNIT=DAY, FMT=20, ERR=1000) IDAY
20
     FORMAT(I2)
      MDAY = IEND(MO) + IDAY
      X=MDAY
С
С
      FIND CLEAR DAY SOLAR RADIATION IN CAL/SQ CM/DAY (CLRAD)
С
           AND CONVERT TO MJ/SO METER/DAY (RSO). THIS IS IN
```

```
С
           CAL/SQ CM/DAY BECAUSE THAT IS HOW COEFFICIENTS WERE
С
           ORIGINALLY DETERMINED FOR EACH STATION IN POLYNOMIAL
С
           EQUATION FROM CLIMAT.DAT.
С
      CLRAD=CLDYC(1)+CLDYC(2)*X+CLDYC(3)*X**2.+
           CLDYC(4)*X**3.+CLDYC(5)*X**4.
      IF (CLRAD.LT.CLDYM) CLRAD=CLDYM
       RSO=CLRAD*.041868
С
С
      SOLVE FOR NET EMITTANCE
С
       A1=.26+.1*EXP(-1.*(.0154*(X-177.))**2.)
        B1 = -0.139
        EMIT=A1+B1*SQRT(ABS(VAPDP))
С
С
      SOLVE FOR NET OUTGOING LONG WAVE RADIATION MJ/SQ M/DAY
С
      RBO=EMIT*(4.903E-9)*(KMAX**4.+KMIN**4.)/2.
С
С
      SOLVE FOR SHORTWAVE (SOLAR) RADIATION REFLECTANCE (ALBEDO)
С
        ALBEDO=.29+.06*SIND(X+97.92)
С
С
        SOLVE FOR CLOUD COVER EXPRESSION
С
      IF (RSO.EQ.0.0) RSO=.000001
      SOL=SOLAR/RSO
      IF (SOL.GT.1.0) SOL=1.0
        IF (SOL.GT.0.70) THEN
          A=1.126
          B = -0.07
        ELSE
          A=1.017
          B = -0.06
        END IF
С
С
        COMPUTE ESTIMATED LONGWAVE (THERMAL) OUTGOING RADIATION
С
        RB=RBO* (A*SOL+B)
С
С
        COMPUTE NET INCOMING RADIATION
С
      RN=(1.-ALBEDO) *SOLAR-RB
С
С
       COMPUTE VAPOR PRESSURE DEFICIT
С
      VAPDE= ( (VAPMN+VAPMX) * .5) -VAPDP
С
С
      COMPUTE WIND FUNCTION (WF) AND COEFFICIENTS (AW) AND (BW)
С
          THESE COEFFICIENTS ASSUME ANEMOMETER HT OF 2 METERS (7 FT OK)
```

```
С
          IF ANY ARE DIFFERENT IN FUTURE, A HEIGHT COVERSION FUNCTION
С
          WILL NEED TO BE ADDED HERE.
С
        AW = .4 + 1.4 \times EXP(-1.*((X-173.)/58.)**2.)
        BW = .007 + .004 \times EXP(-1.*((X-243.)/80.)**2.)
        WF= AW+BW*U2
С
С
      SOLVE FOR EVAPO RATE PENMAN METHOD
С
      ETP LAMB=FACT1*(RN-HTFLUX)+FACT2*6.43*WF*VAPDE
С
С
        CONVERT FROM MJ/SQ M/DAY TO INCHES/DAY
С
      ETP VALUE=ETP LAMB*.03937/PLAMBDA
С
С
        PRINT VALUES TO SCREEN
С
      PRINT *, 'OB MAX', OB MAX
      PRINT *, 'TMAX', TMAX
      PRINT *,'KMAX',KMAX
      PRINT *, 'OB MIN', OB MIN
      PRINT *, 'TMIN', TMIN
      PRINT *,'KMIN',KMIN
      PRINT *, 'OB MEAN', OB MEAN
      PRINT *, 'TMEAN', TMEAN
      PRINT *, 'TP VALUE', TP VALUE
      PRINT *, 'TDEW', TDEW
      PRINT *,'MM SAVE',MM_SAVE
      PRINT *, 'PREAV', PREAV
      PRINT *, 'TPREAV', TPREAV
      PRINT *, 'JLAMBDA', JLAMBDA
      PRINT *, 'PLAMBDA', PLAMBDA
      PRINT *, 'DELTA', DD
      PRINT *, 'GAMMA', GG
      PRINT *, 'PRESSURE', PP
      PRINT *, 'FACT1', FACT1
      PRINT *, 'FACT2', FACT2
      PRINT *, 'SR VALUE', SR VALUE
      PRINT *, 'SOLAR', SOLAR
      PRINT *,'RN',RN
      PRINT *, 'RB', RB
      PRINT *, 'RBO', RBO
      PRINT *, 'CLRAD', CLRAD
      PRINT *, 'RSO', RSO
      PRINT *,'ALBEDO',ALBEDO
      PRINT *,'A1',A1
      PRINT *, 'B1', B1
      PRINT *,'A',A
      PRINT *, 'B', B
      PRINT *, 'DAY', X
```

```
PRINT *,'AW',AW
      PRINT *, 'BW', BW
      PRINT *,'WF',WF
      PRINT *,'WR_VALUE',WR_VALUE
      PRINT *,'U2',U2
      PRINT *,'HTFLUX',HTFLUX
      PRINT *, 'VAPMX', VAPMX
      PRINT *,'VAPMN',VAPMN
      PRINT *,'VAPDP',VAPDP
      PRINT *,'VAPDE',VAPDE
      PRINT *, 'EMIT', EMIT
      PRINT *, 'ET', ETP VALUE
      PRINT *, 'EJ', ETPJ VALUE
      PRINT *, 'CONSTANTS', STN, CLDYC(1), CLDYC(2), CLDYC(3), CLDYC(4)
      1 , CLDYC (5), CLDYM, ANEMH, ELEV, CSUBT, TSUBX
С
      CLOSE (UNIT=76, ERR=1002)
С
      IF(ETP_VALUE.LT.0.0)ETP_VALUE=0.0
      IF(ETPJ VALUE.LT.0.0)ETPJ_VALUE=0.0
С
С
      WRITE ETP AND EJ TO A FILE FOR COMPARISON TO OLD ETP
С
С
      WRITE (15,9800) ETP VALUE, ETPJ VALUE
C9800 FORMAT(1X,2E10.3)
      RETURN
1000 WRITE(6,1001,ERR=925) STATION
1001 FORMAT(' %ETPCOMP-1001-ERROR COMPUTING ET FOR ',A)
925 CLOSE (UNIT=76, ERR=950)
950 RETURN 1
С
1002 RETURN
      END
```

APPENDIX B - Program JENSEN

```
THIS PROGRAM CALCULATES CSUBT AND TSUBX WHICH ARE USED IN JENSENS
С
С
      EQUATION FOR CALCULATING EVAPOTRANSPIRATION POTENTIAL IN THE
С
      IRRIGATION SCHEDULING PROGRAM. INPUT CONSISTS OF LONG TERM MEAN
С
      MAXIMUM AND MINIMUM TEMPERATURES IN DEGREE F. FORTTHE MONTH OF
С
      HIGHEST MEAN AIR TEMPERATURES, AND THE ELEVATION OF THE WEATHER
С
      STATION IN FEET. INPUT IS PROMPTED BY THE PROGRAM.
      PRINT*, 'ENTER MEAN MINIMUM TEMPERATURE'
      READ*, TMIN
      PRINT*, 'ENTER MEAN MAXIMUM TEMPERATURE'
      READ*, TMAX
      PRINT*, 'ENTER ELEVATION IN FEET'
      READ*, ELEV
С
С
      OBTAIN SATURATED VAPOR PRESSURE IN MB FOR THE TWO TEMPS
      TEMPF = TMIN
      I=1
  100 \text{ TEMPC} = (5.0 / 9.0) * (TEMPF - 32.0)
      TEMPK = TEMPC + 273.0
      CA = TEMPK / 273.0
      CB = ALOG(CA)
      CC = CB * 5.31
      CD = (TEMPK - 273.0) / TEMPK
      CE = (CD * 25.22) - CC
      VAPORP = 6.105 * EXP(CE)
      IF(I .EQ. 2) GO TO 110
      VP1 = VAPORP
      TEMPF = TMAX
      I=I+1
      GO TO 100
  110 \text{ VP2} = \text{VAPORP}
      VPDIFF = VP2 - VP1
С
С
      COMPUTE CSUBH , CSUBT , TSUBX VALUES
С
      CONE = 68.0 - (3.6 * ELEV / 1000.0)
      CSUBH = 50.0 / VPDIFF
      CSUBT = 1.0 / (CONE + 13.0 * CSUBH)
      TSUBX = 27.5 - 0.25 * VPDIFF - ELEV / 1000.0
      PRINT 120, VPDIFF, CONE, CSUBH, CSUBT, TSUBX
      STOP
С
С
      FORMATS
С
  120 FORMAT (3X, 28HVAPOR PRESSURE DIFFERENCE = ,F12.5/
     +3X, 7HCONE = ,F12.5/3X, 8HCSUBH = ,F12.5/
     +3X,
     +50H THE FOLLOWING ARE USED IN THE IRRIGATION PROGRAM,/
```

+3X, 14H CSUBT = ,F12.5/ 3X, 14H TSUBX = ,F12.5) END

APPENDIX C - Data File CLIMAT.DAT

STN is the weather station code

CLDYC(1) thru CLDYC(5) are coefficients used in a fifth degree polynomial along with the day of the year to compute clear day solar radiation. CLDYM is the clear day minimum solar radiation (cal/cm 2 /d) AHT is the anemometer height in feet

ELEV is the weather station site elevation in meters above mean sea level CsubT is used in the Jensen Haise equation

 ${\tt TsubX}$ is used in the Jensen Haise equation

Data File: CLIMAT.DAT

STN	CLDYC1	CLDYC2	CLDYC3	CLDYC4	CLDYC5	CLDYM	AHT	ELEV	CsubT 5	<u> [subX</u>
LIDW	.154E+3	.176E+1	.614E-1	396E-3	.590E-6	100.0	7.0	497.0	0.0115	19.3
HERO	.183E+3	.151E+1	.611E-1	389E-3	.578E-6	100.0	7.0	199.0	0.0115	19.2
MALI	.213E+3	.960E+0	.599E-1	362E-3	.521E-6	100.0	7.0	1345.0	0.0139	15.0
ODSW	.154E+3	.176E+1	.614E-1	396E-3	.590E-6	100.0	7.0	503.0	0.0113	19.8
MRSO	.195E+3	.703E+0	.680E-1	402E-3	.575E-6	100.0	7.0	744.0	0.0114	19.3
COVM	.165E+3	.187E+1	.584E-1	381E-3	.596E-6	100.0	7.0	1096.0	0.0122	17.9
OMAW	.132E+3	.144E+1	.684E-1	432E-3	.643E-6	100.0	7.0	376.0	0.0114	19.4
CHVO	.199E+3	.231E+1	.476E-1	326E-3	.490E-6	100.0	7.0	1329.0	0.0121	17.8
CEDC	.219E+3	.245E+1	.437E-1	305E-3	.460E-6	100.0	7.0	1402.0	0.0124	17.4
BANO	.201E+3	.231E+1	.479E-1	328E-3	.493E-6	100.0	7.0	24.0	0.0061	25.7
PMAI	.193E+3	.218E+1	.508E-1	342E-3	.514E-6	100.0	7.0	703.0	0.0122	18.0
GERW	.147E+3	.191E+1	.601E-1	392E-3	.587E-6	100.0	7.0	351.0	0.0111	20.2
GOLW	.170E+3	.164E+1	.610E-1	391E-3	.582E-6	100.0	7.0	512.0	0.0113	19.6
LEGW	.179E+3	.148E+1	.620E-1	394E-3	.585E-6	100.0	7.0	259.0	0.0111	20.2
HOXO	.167E+3	.194E+1	.573E-1	376E-3	.562E-6	100.0	7.0	171.0	0.0100	22.2
HRHW	.170E+3	.164E+1	.610E-1	391E-3	.582E-6	100.0	7.0	274.0	0.0115	19.4
AHTI	.191E+3	.217E+1	.512E-1	344E-3	.516E-6	100.0	7.0	1615.0	0.0131	16.2
RXGI	.191E+3	.217E+1	.512E-1	344E-3	.516E-6	100.0	7.0	1486.0	0.0131	6.3
FAFI	.199E+3	.231E+1	.476E-1	326E-3	.490E-6	100.0	7.0	1536.0	0.0139	15.1
AFTY	.204E+3	.228E+1	.480E-1	328E-3	.492E-6		7.0		0.0147	
RPTI	.207E+3	.238E+1	.461E-1	319E-3	.480E-6		7.0		0.0137	
ECHO	.183E+3	.151E+1	.611E-1	389E-3	.578E-6	100.0	7.0	232.0	0.0116	19.0
HRMO	.183E+3	.151E+1	.611E-1	389E-3	.578E-6	100.0	7.0		0.0116	19.0
LAKO	.219E+3	.245E+1	.437E-1	305E-3	.460E-6	100.0	7.0	1436.0	0.0135	15.7
CRSM	.124E+3	.203E+1	.614E-1	403E-3	.605E-6	100.0	7.0	899.0	0.0121	18.2
RDBM	.124E+3	.203E+1	.614E-1	403E-3	.605E-6	100.0	7.0	927.0	0.0121	18.1
PCYO	.195E+3	.703E+0	.680E-1	402E-3	.575E-6	100.0	7.0	1144.0	0.0121	18.0
MDFO	.219E+3	.245E+1	.437E-1	305E-3	.460E-6	100.0	7.0	408.0	0.0116	19.0
CRVO	.185E+3	.208E+1	.531E-1	354E-3	.530E-6	100.0	7.0	70.0	0.0105	21.4
TWFI	.207E+3	.240E+1	.459E-1	318E-3	.478E-6	100.0	7.0	1194.0	0.0129	16.6
ABEI	.204E+3	.228E+1	.480E-1	328E-3	.493E-6	100.0	7.0	1332.0	0.0130	16.5
SIGM	.155E+3	.160E+1	.636E-1	406E-3	.606E-6	100.0	7.0	895.5	0.0121	18.1
FOGO	.190E+3	.104E+1	.629E-1	378E-3	.543E-6		7.0		0.0108	
ONTO	.193E+3	.218E+1	.508E-1	342E-3	.514E-6	100.0	7.0	688.0	0.0122	18.0

```
.141E+1 .557E-1 -.355E-3 .525E-6 100.0 7.0 786.0 0.0122 18.0
GDVI .234E+3
            .228E+1 .480E-1 -.328E-3 .493E-6 100.0 7.0 1354.0 0.0130 16.5
FTHI .204E+3
GFRI .234E+3
                      .557E-1 -.355E-3 .525E-6 100.0 7.0 921.4 0.0122 18.0
             .141E+1
                                        .490E-6 100.0 7.0 1485.0 0.0139 15.1
PICI .199E+3
                              -.326E-3
             .231E+1
                      .476E-1
POBO .255E+3
             .139E+1
                      .695E-1
                              -.408E-3 .592E-6 100.0 7.0 974.7 0.0125 17.5
MASW .141E+3
             .156E+1
                     .655E-1
                              -.417E-3 .622E-6 100.0 6.6 600.6 0.0120 18.3
                      .602E-1 -.392E-3 .590E-6 100.0 6.6 837.6 0.0124 17.5
IMBO .180E+3
             .169E+1
             .218E+1
                              -.342E-3
                                         .514E-6 100.0 6.6 828.5 0.0122 18.0
BOII .193E+3
                      .508E-1
NMPI .193E+3
                              -.342E-3 .514E-6 100.0 6.6 802.8 0.0122 18.0
             .218E+1
                      .508E-1
                      .519E-1 -.348E-3 .522E-6 100.0 6.6 1158.0 0.0124 17.5
HRFO .189E+3
             .213E+1
                              -.392E-3
                                        .584E-6 100.0 6.6 1418.0 0.0124 17.5
DRLM .161E+3
             .175E+1
                      .606E-1
             .134E+1
                      .657E-1
                              -.412E-3
                                        .611E-6 100.0 6.6
ARAO .182E+3
                                                          42.4 0.0108 21.0
                      .450E-1 -.312E-3 .470E-6 100.0 6.6 1249.7 0.0135 15.7
KFLO .215E+3
             .240E+1
BRKO .213E+3
             .242E+1
                      .449E-1 -.312E-3 .470E-6 100.0 6.6
                                                            24.0 0.0061 25.7
WRDO .215E+3
             .240E+1
                      .450E-1
                              -.312E-3
                                        .470E-6 100.0 6.6 1249.7 0.0135 15.7
AGKO .215E+3
             .240E+1
                              -.312E-3
                                        .470E-6 100.0 6.6 1249.7 0.0135 15.7
                      .450E-1
LORO .215E+3
             .240E+1
                      .450E-1 -.312E-3
                                        .470E-6 100.0 6.6 1268.0 0.0135 15.7
                      .369E-1 -.270E-3
                                        .409E-6 100.0 6.6 1208.5 0.0000 00.0
FALN .246E+3
             .269E+1
                              -.373E-3
                                        .558E-6 100.0 6.6 1042.4 0.0000 00.0
BKVO .171E+3
             .195E+1
                      .568E-1
EURN .246E+3
             .269E+1
                      .369E-1
                              -.270E-3 .409E-6 100.0 6.6 1797.4 0.0000 00.0
                              -.374E-3 .559E-6 100.0 6.6
                                                           24.1 0.0000 00.0
BNDW .168E+3
             .196E+1
                     .569E-1
                              -.418E-3 .623E-6 100.0 6.6 304.0 0.0000 00.0
CJDW .143E+3
             .153E+1
                      .659E-1
             .161E+1
GCDW .142E+3
                      .649E-1
                              -.414E-3
                                        .617E-6 100.0 6.6 404.0 0.0000 00.0
SBMW .143E+3
             .162E+1
                      .646E-1 -.413E-3
                                        .616E-6 100.0 6.6 419.0 0.0000 00.0
KFLW .134E+3
             .147E+1
                      .676E-1 -.427E-3
                                        .636E-6 100.0 6.6 407.0 0.0000 00.0
                              -.394E-3
                                         .588E-6 100.0 6.6 192.0 0.0000 00.0
LBRW .156E+3
             .179E+1
                      .608E-1
                              -.393E-3
                                        .587E-6 100.0 6.6 506.0 0.0000 00.0
DENI .158E+3
             .177E+1
                      .608E-1
SILW .160E+3
             .180E+1
                      .600E-1
                              -.389E-3
                                        .581E-6 100.0 6.6 251.0 0.0000 00.0
DTRO .178E+3
                      .538E-1 -.358E-3
                                        .537E-6 100.0 6.6 502.9 0.0000 00.0
             .210E+1
HCKO .192E+3
             .222E+1
                      .507E-1
                              -.342E-3
                                         .513E-6 100.0 6.6 438.9 0.0000 00.0
LKPO .192E+3
                              -.342E-3
                                        .513E-6 100.0 6.6 283.5 0.0000 00.0
             .222E+1
                      .507E-1
                              -.348E-3
                                        .522E-6 100.0 6.6 1104.9 0.0000 00.0
BEWO .188E+3
             .215E+1
                      .519E-1
                               -.376E-3
                                        .563E-6 100.0 6.6 189.0 0.0000 00.0
PNGO .169E+3
             .192E+1
                      .574E-1
                              -.375E-3 .561E-6 100.0 6.6 384.0 0.0000 00.0
DEFO .170E+3
             .192E+1
                      .572E-1
BATO .206E+3 .241E+1
                     .464E-1 -.321E-3 .482E-6 100.0 6.6 1317.0 0.0000 00.0
CHAW .141E+3 .158E+1 .653E-1 -.416E-3 .620E-6 100.0 6.6 594.0 0.0000 00.0
```

REFERENCES

Am. Soc. Agric. Engrs. (1980). <u>Design and Operation of Farm Irrigation Systems</u>, Am. Soc. Agric. Engrs. Mongraph No. 3., 829 p., pgs. 195-197.

Am. Soc. Civil Engrs. (1990). <u>Evapotranspiration and Irrigation Water Requirements</u>, Am. Soc. Civil Engrs. Manuals and Reports on Engineering Practice, Publ. No. 70., 332 p., pgs. 25-35,38,132-140,168,169,174,175.

Hill, Dr. R.W., Johns, E.L., and Frevert, D.K., (1983). "Comparison of Equations Used For Estimating Crop Evapotranspiration With Field Research." Dep. Rep., U.S. Bureau of Reclamation. Denver, Co. pgs. 2-6, 2-9 thru 2-14.

McVay, M., U.S. Bureau of Reclamation, Boise, Idaho, (1992). "AgriMet - An Automatic Agricultural Weather Monitoring System", written for presentation at the American Society of Agricultural Engineers, Pacific Northwest Section, 47th Annual Meeting, Bozeman, Montana, September 16-18, 1992.

Powers, A., U.S. Bureau of Reclamation, Boise, Idaho, (1992). "Crop Water Use Modeling Through Satellite Telemetry", written for presentation at the American Society of Agricultural Engineers International Summer Meeting in Charlotte, North Carolina, June 21-24, 1992.